
1. Introduzione

OLE for Process Control (OPC) fornisce un meccanismo affinché una sorgente dati comunichi con qualsiasi applicazione client in modo standard. È così possibile per i fornitori hardware offrire ai propri acquirenti dei server con interfaccia OPC, permettendo a qualsiasi client di accedere alle loro apparecchiature. 

OPC prevede due tipi di interfacce:

· Custom Interfaces
· Automation Interfaces
L’interfaccia Custom è utlizzata in linguaggi che prevedono l’utilizzo di puntatori (C,C++), si tratta dunque di una programmazione abbastanza complessa.

L’interfaccia Automation invece è utilizzata in linguaggi come il Visual Basic che non prevedono l’utilizzo di puntatori, rendendo così la programmazione più semplice senza però sacrificare le funzionalità definite dall’ Interfaccia Custom.

Ciò è possibile grazie ad un ‘wrapper’ DLL che funge da tramite tra un Automation Client ed un OPC Custom Interface Server, come mostrato in figura

L’insieme delle variabili (Item) che un server espone prende il nome di Namespace.

Un client può utilizzare anche solo una parte degli item esposti dal server e per farlo deve stanziare una serie di oggetti rispecchiando la gerarchia del modello dell’oggetto OPC Autamation Server rappresentato di seguito

OPCServer
È l’istanza di un OPC Server. Bisogna creare un oggetto OPCServer prima di fare riferimento a qualunque altro oggetto.

OPCGroups
È una collezione di oggetti OPCGroup che il client crea all’interno di un OPCServer.

OPCGroup
È l’istanza di un oggetto OPCGroups; lo scopo di questo oggetto è quello di mantenere informazioni di stato e fornire un meccanismo che consenta ad un oggetto OPCItems istanziato da un oggetto OPCGroup di acquisire dati.

OPCItems
E’ una collezione di tutti gli oggetti OPCItem che il client ha creato all’interno dell’OPCServer e all’interno dell’OPCGroup.

OPCItem
Oggetto che contiene   Valore, Informazioni di stato, Tempo dell’ultima creazione di un item. 

Da notare che l’ interfaccia Custom non prevede un oggetto Item separato.

OPCBrowser
È un oggetto che permette il browsing degli item del server OPC a cui si è connessi. Esiste un solo OPCBrowser per ogni istanza di un oggetto ServerOPC.

L’OPC Client che ho realizzato è stato implementato in Visual Basic, rifacendomi all’interfaccia standard OPC Data Access Automation. 

Per l’implementazione è stato dunque necessario caricare nei Riferimenti di Visual Basic la libreria OPC Automation 2.0, attraverso la quale si avranno a disposizione le seguenti classi

·  OPCServer
·  OPCGroups

·  OPCGroup

·  OPCItems

·  OPCItem

·  OPCBrowser

Ognuna di esse possiede Metodi e Proprietà e le classi OPCServer, OPCGroups e OPCGroup possiedono anche Eventi.

Per gestire gli eventi, le variabili in Visual Basic devono essere dichiarate con la parola chiave WithEvents

Esempio:

       Public WithEvents Server as OpcServer

Di seguito verrà descritto il funzionamento del Opc Client che ho realizzato, nonché la spiegazione del codice Visual Basic riguardante la libreria OPC Automation 2.0.

2. OPC Client

L’ OPC Client che è stato realizzato prevede la connessione a molteplici Server.

Se la classe OPCServer non avesse avuto Eventi sarebbe bastato creare un array di oggetti OPCServer, ma poiché contiene l’evento ServerShutDown ho dovuto creare una classe chiamata OPCServers.

Questa classe rappresenta una collezione di oggetti OPCServer ed è semplicemente costituita da una variabile di tipo OPCServer e dal relativo evento come mostrato di seguito

Classe OPCServers

Public WithEvents Server As OpcServer

Private Sub Class_Initialize()

   Set server = New OpcServer

End Sub

Private Sub Server_ServerShutDown(ByVal Reason As String)

   istruzioni

End Sub

Quando viene creato un oggetto OPCServers, attraverso la procedura Class_Initialize() viene automaticamente creato un oggetto OPCServer il cui riferimento è associato alla variabile Server.

La variabile Server è pubblica e costituisce l’unica proprietà della classe OPCServers, inoltre poiché è dichiarata come WithEvents, gestisce gli eventi della classe OPCServer che possiede un solo evento e precisamente l’evento ServerShutDown.

Poiché è possibile creare più gruppi anche per uno stesso server e poiché la classe OPCGroup contiene diversi eventi, anche in questo caso è stato necessario creare una classe che ho chiamato OPCGruppi.

Classe OPCGruppi

Public WithEvents group As OPCGroup

Public NomeServer As String

Private Sub Class_Initialize()

      Set group = ServerConnessi(FormCreaGruppo.NomeServer.ListIndex + 1).server      

                          .OpcGroups(ServerConnessi(FormCreaGruppo.NomeServer.ListIndex + 1)   

                          .server.OpcGroups.Count)

End Sub

Private Sub group_AsyncReadComplete(ByVal TransactionID As Long, ByVal NumItems  

                                                                 As Long,ClientHandles() As Long, ItemValues() 

                                                                 As Variant, Qualities() As Long,TimeStamps() As 

                                                                 Date, Errors() As Long)

      istruzioni

End Sub

Private Sub group_DataChange(ByVal TransactionID As Long, ByVal NumItems As Long, 

                                                   ClientHandles() As Long,ItemValues() As Variant,     

                                                   Qualities() As Long, TimeStamps() As Date)

      istruzioni

End Sub

La proprietà NomeServer contiene il nome del server a cui il gruppo appartiene.

Quando viene creato un oggetto OPCGruppi, attraverso la procedura Class_Initialize() viene automaticamente creato un oggetto OPCGroup il cui riferimento è associato alla variabile group.                                                                        Poiché la variabile group è dichiarata come WithEvents gestisce gli eventi della classe OPCGroup che sono 

· AsyncReadComplete

· AsyncWriteComplete

· AsyncCancelComplete

· DataChange

Oltre ai due Moduli di classe sopra citati è presente anche un Modulo che ho chiamato “Public” contenente variabili, procedure e funzioni con scope globale, tra queste ci sono le seguenti dichiarazioni di variabili

Public ServerConnessi() As OPCServers

Public CollezioneGruppi() As OPCGruppi

La variabile pubblica ServerConnessi() è un array dinamico di oggetti OPCServers, ogni qualvolta viene effettuata una connessione ad un server la sua dimensione viene aumentata di una unità, per accedere all’oggetto OPCServer della libreria OPC Automation 2.0 basta scrivere

    ServerConnessi(index).Server

La variabile pubblica CollezioneGruppi() è un array dinamico di oggetti OPCGruppi, ogni qualvolta viene creato un gruppo, sia dello stesso server che di server diversi, la sua dimensione viene aumentata di una unità, per accedere all’oggetto OPCGroup della libreria OPC Automation 2.0 basta scrivere

    CollezioneGruppi(index).group

A questo punto vediamo le caratteristiche del OPC Client.

Avviando il programma apparirà la seguente schermata

La prima cosa da fare è stabilire una connessione ad un server, per farlo premere il pulsante Connessione Server nella Barra Pulsanti oppure scegliere dal menù a tendina OPC Server e quindi Connetti.

Apparirà la seguente Finestra

Nella casella di testo IP Server bisogna immettere il nome del computer del quale si vuole visualizzare l’elenco dei server presenti e premere Aggiorna lista Opc Server. 

Se IP Server è vuoto vengono visualizzati i server presenti nella Macchina Locale

Premendo il pulsante Aggiorna lista Opc Server viene invocato il metodo GetOpcServers(), di seguito il codice riguardante

Dim server As OpcServer

Dim ArrayOPCServer As Variant

Set server = New OpcServer

ArrayOPCServer = server.GetOPCServers(IPServer.Text)

Il metodo GetOPCServers() appartiene alla classe OPCServer, per cui è necessario prima creare un’istanza della classe OPCServer.

Nel codice il riferimento a tale istanza è assegnata alla variabile Server.

A questo punto è possibile richiamare il metodo GetOPCServers() che vuole come parametro l’indirizzo IP della macchina di cui si vuole ottenere la lista dei server, che nel nostro caso è dato dal valore immesso nella casella di testo IP Server a cui si accede con l’istruzione IPServer.text.

Il metodo GetOPCServers() torna un array di Variant dei nomi dei server registrati. 

Una volta selezionato il server a cui ci si vuole connettere premere il pulsante Connetti, di seguito viene descritto parte del codice che viene eseguito

ReDim Preserve ServerConnessi(UBound(ServerConnessi) + 1)

Set ServerConnessi(UBound(ServerConnessi)) = New OPCServers

ServerConnessi(UBound(ServerConnessi)).server.Connect ListOpcServer.List(ListOpcServer.ListIndex), IPServer.Text

La prima istruzione aumenta di una unità la dimensione dell’array ServerConnessi(), che ricordo è dichiarato nel modulo Public.

La seconda istruzione crea un oggetto OPCServers e ne associa il riferimento all’ultimo elemento dell’array ServerConnessi(), ricordo che quando viene creato un oggetto OPCServers viene automaticamente creato un oggetto OPCServer a cui si accede attraverso la proprietà Server dell’oggetto OPCServers.

La terza istruzione, infine, effettua la connessione al server specificato attraverso il metodo connect() dell’oggetto OPCServer. 

Il metodo connect() vuole come parametri il nome del server che, nel nostro caso, è restituito dall’istruzione ListOpcServer.List(ListOpcServer.ListIndex) e l’indirizzo IP del computer in cui si trova il server che, nel nostro caso, è restituito dall’istruzione IPServer.text.

Se la connessione và a buon fine apparirà la seguente schermata

Nel TreeView comparirà il nome del  server a cui si è effettuata la connessione, che in questo esempio è il server Matrikon.OPC.Simulation.

A destra del TreeView (ridimensionabile)  ci sono due elenchi, in quello superiore verranno visualizzati i gruppi con i relativi attributi ed in quello inferiore gli item anch’essi con relativi attributi.

A questo punto dobbiamo creare un gruppo, per farlo premere nella Barra Pulsanti il pulsante Crea Gruppo oppure selezionare OPC Groups dal menù a tendina e quindi Crea, apparirà la seguente finestra

In Nome Server và selezionato il server in cui si vuole aggiungere il gruppo, all’avvio comparirà il nome del server selezionato nel TreeView.

In Nome Gruppo và inserito il nome che si desidera assegnare al gruppo.

All’apertura della finestra, le proprietà Time Bias, Dead Band, Update Rate e Is Active assumeranno rispettivamente i valori di DefaultTimeBias, DefaultDeadBand, DefaultUpdateRate e DeafultIsActive che sono proprietà dell’oggetto OPCGroups e possono essere modificate in ogni momento selezionando dal menù a tendina OPCGroups e quindi Proprietà di Default.
Gli altri campi settano gli attributi del gruppo:

· TimeBias


Long contenente il TimeBias iniziale (in minuti) per il Group.

· Dead Band


L’intervallo di valori per la deadband va dallo 0% al 100%. La  


deadband si applica solo agli item di tipo analogici che possiedono le  


proprietà EU Low ed EU High. Questi due valori servono per verificare        


la seguente condizione 

                       abs(last cached value - current value) > (Deadband/100) * (EU High - EU Low) 

Solo se tale condizione è verificata l’ultimo valore contenuto nella cache  


verrà aggiornato

· UpdateRate


Rappresenta la frequenza con cui vengono aggiornati i dati nella cache

· IsActive


Se questa proprietà è impostata su false allora tutti gli item che fanno parte 
del gruppo non verranno aggiornati nella cache, ne segue che una lettura 
sincrona dalla cache restituirà una quality bassa e il refresh del gruppo   

         non verrà eseguito

· IsSubscribed


Se questa proprietà è impostata su false il server non notificherà eventuali 
letture a scritture asincrone, ne segue che oltre a non essere effettuate 
letture e scritture asincrone, non verrà neanche generato l’evento 
DataChange  

Modificate opportunamente le proprietà e scelto il nome del gruppo (in questo esempio Gruppo1), premere il pulsante Crea Gruppo, parte del codice che verrà eseguito è il seguente

‘Crea il Gruppo 

ServerConnessi(NomeServer.ListIndex + 1).server.OPCGroups.Add NomeGruppo.Text   ‘Aumenta di una unità la dimensione dell’array CollezioneGruppi

ReDim Preserve CollezioneGruppi(UBound(CollezioneGruppi) + 1) 

‘Crea un oggetto OPCGruppi

Set CollezioneGruppi(UBound(CollezioneGruppi)) = New OPCGruppi   

‘Assegna alla proprietà NomeServer dell’oggetto OPCGroup il nome del server

CollezioneGruppi(UBound(CollezioneGruppi)).NomeServer = ServerConnessi(NomeServer.ListIndex + 1).server.ServerName

If CheckIsActive.Value = 1 Then   ‘controlla se la casella Is Active è selezionata
   CollezioneGruppi(UBound(CollezioneGruppi)).group.IsActive = True

Else

   CollezioneGruppi(UBound(CollezioneGruppi)).group.IsActive = False

End If

If CheckIsSubscribed.Value = 1 Then    ‘controlla se la casella Is Subscribed è selezionata

   CollezioneGruppi(UBound(CollezioneGruppi)).group.IsSubscribed = True

Else

   CollezioneGruppi(UBound(CollezioneGruppi)).group.IsSubscribed = False

End If

CollezioneGruppi(UBound(CollezioneGruppi)).group.DeadBand = DeadBand.Text

CollezioneGruppi(UBound(CollezioneGruppi)).group.TimeBias = TimeBias.Text

CollezioneGruppi(UBound(CollezioneGruppi)).group.UpdateRate = UpdateRate.Text

La prima istruzione crea il gruppo utilizzando il metodo Add() dell’oggetto OPCGroups, tale metodo ha come unico parametro il nome del gruppo che, nel nostro caso, è restituito da NomeGruppo.Text

Attraverso l’istruzione CollezioneGruppi(UBound(CollezioneGruppi)).group si accedono a tutte le proprietà dell’oggetto OPCGroup appena creato e le istruzioni successive ne settano i valori con quelli indicati nella finestra Crea Gruppo

Se il metodo Add() và a buon fine nell’elenco dei gruppi apparirà quello appena creato con i relativi attributi, che possono essere modificati in ogni momento selezionando il gruppo (i gruppi selezionati hanno uno sfondo rosso) e premendo nella Barra Pulsanti Modifica Gruppo.

Per selezionare un gruppo cliccare sulla riga corrispondente

Per modificare un gruppo è necessario selezionare soltanto un gruppo, per altre operazioni è possibile effettuare una selezione multipla attraverso i tasti SHIFT e CTRL

Per aggiungere gli item premere il pulsante Aggiungi Item nella Barra Pulsanti oppure selezionare OPC Items e quindi Aggiungi dal menù a tendina, apparirà la seguente finestra

In ciascun server gli item possono essere organizzati in modo flat o gerarchico.  Nel treeview della finestra Aggiungi Item appariranno tutti i rami e affianco, nel listview, tutti gli item del ramo selezionato, questo è possibile grazie alla classe OPCBrowser. 

Di seguito è riportato parte del codice che viene eseguito all’apertura della finestra Aggiungi Item

Dim i As Integer

Dim branches() As String

‘Ciclo for esteso a tutti i server connessi

For i = 1 To UBound(ServerConnessi)   

   ‘Viene creato un oggetto OPCBrowser il cui riferimento è associato alla variabile browser

   ‘La variabile browser è dichiarata a livello di form ed è dunque visibile da tutti gli oggetti   

   ‘che il form Aggiungi Item contiene

   Set browser = ServerConnessi(i).server.CreateBrowser   

   browser.MoveToRoot   ‘permette di spostarsi al primo livello dell’albero
   ReDim branches(0)   ‘Resetta la dimensione dell’array dinamico branches
   branches(0) = Empty   ‘Imposta ad empty il valore dell’elemento 0
   ‘Richiama la procedura RiempiTreeView

   RiempiTreeView ServerConnessi(i).server.ServerName, branches   

Next i

La variabile branches è un array dinamico che contiene, attraverso i suoi elementi, il percorso del livello corrente nell’albero, questo array insieme al nome del server vengono passati come parametri alla procedura  RiempiTreeView.

La procedura RiempiTreeView verrà richiamata in modo ricorsivo al fine di sfogliare l’intero albero.

In questa prima chiamata poiché lo stato corrente è il primo livello, l’array branches non conterrà alcun elemento.

Di seguito è riportato il codice della procedura RiempiTreeView
Private Sub RiempiTreeView(NomeItem As String, branches() As String)

Dim i, numerobranches As Long

‘ Controlla che il livello corrente dell’albero non sia il primo

If UBound(branches) <> 0 Then   

   browser.MoveDown (NomeItem)    ‘ Entra nel ramo specificato da NomeItem
End If

‘ Riempie la collezione con i nomi dei rami della posizione corrente

browser.ShowBranches   

‘ Controlla che la collezione dei nomi dei rami non abbia dimensione nulla

If browser.Count <> 0 Then   

   ‘ Aumenta di una unità la dimensione dell’array branches

   ReDim Preserve branches(UBound(branches) + 1)   

   ‘ Se la dimensione dell’array branches è uguale ad 1 vuol dire che il livello corrente è il    

   ‘primo, ma nell’oggetto OPCBrowser il primo livello si indica con “”  e i successivi invece    

   ‘con i nomi dei rami, il seguente If  distingue i due casi

   If UBound(branches) = 1 Then                                 

      branches(UBound(branches)) = "“                       

   Else                                                                             

      branches(UBound(branches)) = NomeItem         

   End If

   ‘ Nel successivo ciclo for ci si sposta in diverse posizioni dell’albero per cui è                                                     

   ‘ necessario  assegnare ad una variabile il numero di rami del livello corrente

   numerobranches = browser.Count  

   ‘ Esegue  le istruzioni all’interno del ciclo for per tutti i rami del livello corrente

   For i = 1 To numerobranches    

      ‘ Entra nel ramo specificato dal percorso contenuto in branches

      browser.MoveTo branches    

      ‘ Riempie la collezione con i nomi dei rami della posizione corrente

      browser.ShowBranches    

      ‘ Aggiunge il ramo nel TreeView

      TreeView.Nodes.Add NomeItem, tvwChild, browser.Item(i), browser.Item(i), 2, 3                 

      ‘ Richiama ricorsivamente la procedura RiempiTreeView

      RiempiTreeView browser.Item(i), branches 

      ‘ Il seguente If controlla se si è arrivati all’ultimo elemento della collezione dei rami   

      ‘perché in tal caso la dimensione dell’ array branches dovrà essere diminuito di una unità

      If i = numerobranches Then                                             

         ReDim Preserve branches(UBound(branches) - 1)     

      End If                                                                                  

   Next i

End If

End Sub

Quando si seleziona un ramo del TreeView, nel ListView appariranno gli item del ramo selezionato secondo i criteri specificati in Filtro, in particolare è possibile scegliere se visualizzare solo gli item che hanno come iniziali del nome quelli specificati in Nome Item oppure solo quelli con un specifico tipo di dato o ancora quelli che possono essere letti, scritti o letti e scritti.                                                      Di seguito parte del codice che viene eseguito quando si seleziona un ramo nel TreeView

‘ branches contiene il percorso del ramo selezionato e attraverso il metodo MoveTo dell’oggetto   

‘ browser ci si sposta proprio in tale ramo

browser.MoveTo branches   

‘ Se la casella Nome Item è selezionata l’attributo Filter dell’oggetto browser viene impostato 
‘ col nome indicato nella casella di testo Nome Item, con il simbolo “*” vengono visualizzati anche  

‘ gli items che hanno le iniziali indicati in Nome Item
If CheckNomeItem.Value = 1 Then           

     browser.Filter = TextNomeItem & "*“   

end if                                                           

‘ Se la casella Access Rights è selezionata l’attributo AccessRights dell’oggetto browser viene         ‘ settato con il valore indicato nella casella combinata.

‘ OPCReadable ed OPCWritable sono delle costanti della libreria OPC Automation 2.0
If CheckAccessRight.Value = 1

     If ComboAccessRight.Text = "Readable" Then           

          browser.AccessRights = OPCReadable                     

     ElseIf ComboAccessRight.Text = "Writable" Then      

          browser.AccessRights = OPCWritable

     End If

End If

‘Se la casella DataType è selezionata l’attributo DataType dell’oggetto browser viene impostato con ‘il tipo di dato indicato nella casella combinata. 

‘Poiché browser.datatype vuole il codice del tipo di dato e non la stringa come indicato nella ‘casella combinata, ho creato una funzione (CodeDataType) dichiarata nel modulo Public che ‘restituisce il codice del ‘tipo di dato fornendogli come parametro il tipo di dato espresso in stringa

If CheckDataType.Value = 1 Then                                             

     browser.DataType = CodeDataType(ComboDataType.Text)    

End If                                                                                                  

‘ Il metodo showleafs dell’oggetto browser riempie la collezione con i nomi degli item del ramo ‘corrente

browser.ShowLeafs    

‘ Aggiunge al ListView  i nomi degli item del ramo selezionato

‘ I nomi degli item sono restituiti dall’istruzione browser.item(i)

For i = 1 To browser.Count       

     ListaItem.ListItems.Add , , browser.Item(i), , 1     

Next i    

Ogni volta che si seleziona un item del listview, in basso vengono visualizzate le relative proprietà e questo senza aver ancora aggiunto l’item in un gruppo. Di seguito è riportato parte del codice che viene eseguito quando si seleziona un item del listview               

Dim OPCItemID, Value As String

Dim ItemCount As Long

Dim PropertyIDs() As Long

Dim Data() As Variant

Dim Errors() As Long

Dim Descriptions() As String

Dim DataTypes() As Integer

OPCItemID = browser.GetItemID(Item)

ServerConnessi(IndiceServer).server.QueryAvailableProperties OPCItemID, ItemCount,   

                                                                                                       PropertyIDs, Descriptions, 

                                                                                                       DataTypes

ServerConnessi(IndiceServer).server.GetItemProperties OPCItemID, ItemCount, 

                                                                                          PropertyIDs, Data, Errors

Attraverso il metodo GetItemID() dell’oggetto browser è possibile risalire al ItemID dell’item, per far questo però la posizione corrente dell’oggetto browser si deve trovare nel ramo in cui si trova l’item, fatto ciò è sufficiente passare al metodo GetItemID() il nome dell’item.

Ogni proprietà di un item ha un identificativo che sono accessibili grazie al metodo QueryAvailableProperties() dell’oggetto server. Per utilizzare tale metodo bisogna specificare nei primi due parametri rispettivamente l’itemID ed il numero di attributi di cui si vogliono conoscere gli identificativi; gli altri tre parametri verranno utilizzati dal metodo per restituire rispettivamente un array di identificatori, un array di descrizioni ed un array dei tipi dato con cui verrà restituita ciascuna proprietà.

Nei primi tre parametri del metodo GetItemProperties() và indicato rispettivamente l’ItemID, il numero di attributi che si vogliono conoscere e un array contenente gli identificativi degli attributi; gli altri due parametri saranno utilizzati dal metodo per restituire rispettivamente un array dei valori degli attributi ed un array di errori per verificare se ogni proprietà è stata restituita.

All’apertura della finestra Aggiungi Item, nel TreeView sarà selezionato il server in cui si trova il gruppo selezionato nel TreeView della schermata principale e nella casella combinata Gruppo il gruppo selezionato.

Successivamente è possibile spostarsi tra i vari server e gruppi e ad ogni selezione nel TreeView verrà aggiornato, a destra della finestra, il nome del server corrente e l’elenco dei gruppi del server nella casella combinata.

Le proprietà RequestedDataType ed IsActive assumeranno all’apertura della finestra i valori memorizzati rispettivamente dagli attributi DefaultRequestedDataType e DefaultIsActive dell’oggetto OPCItems corrente, questi valori possono essere modificati selezionando dal menù a tendina OPCItems e quindi Proprietà Di Default.

Prima di aggiungere l’item è possibile scegliere se l’item sarà attivo o meno selezionando o no la casella Is Active ed è possibile scegliere il Requested Data Type dalla casella combinata, se si seleziona Empty il RequestedDataType coinciderà con il CanonicalDataType.

La proprietà CanonicalDataType della classe OPCItem contiene il tipo di dato originario dell’item nel server.

La proprietà RequestedDataType della classe OPCItem và settato con il tipo di dato in cui si desidera che il server restituisca il valore dell’item. 

Se la proprietà IsActive delle classe OPCItem è impostata su false allora l’item in esame non verrà aggiornato nella cache, ne segue che una lettura sincrona dalla cache restituirà una quality bassa e il refresh del gruppo che lo contiene non sortirà alcun effetto su tale item, ma aggiornerà tutti gli altri item che sono attivi.

E’ possibile selezionare anche più item, in tal caso però la casella combinata Requested Data type verrà automaticamente impostata su Empty.

A questo punto per aggiungere l’item o gli items selezionati basta premere il pulsante Aggiungi Item, parte del codice che verrà eseguito è il seguente

Dim Items As OpcItems

‘ Assegna alla variabile Items il riferimento all’oggetto OPCItems del gruppo selezionato      ‘ nella casella combinata. Da notare che per ottenere il riferimento al gruppo selezionato ho              ‘ utilizzato il metodo GetOPCGroup dell’oggetto OPCGroups che vuole come parametro       ‘ semplicemente il nome del gruppo

Set Items=ServerConnessi(i).server.OpcGroups.GetOPCGroup(ListaGruppi.Text).OpcItems

‘ Aggiunge l’item selezionato

Items.AddItem browser.GetItemID(ListaItem.ListItems(j).Text), 

                         FormElencoItem.GridItems.Rows

‘ Viene settata la proprietà RequestedDataType dell’oggetto OPCItem appena creato a cui si ‘ accede con l’istruzione Items(indice dell’item, in questo caso l’ultimo).

‘ CodeDataType, come già visto in precedenza, restituisce l’identificativo dell’tipo di dato   ‘ espresso in stringa

Items(Items.Count).RequestedDataType = CodeDataType(ComboRequestedDataType.Text)

‘ Imposta su true o false la proprietà IsActive dell’ultimo oggetto OPCItem, quello appena   ‘ creato, rispettivamente se la casella Is Active è selezionata o meno  

If CheckIsActive.Value = 1 Then                 

     Items(Items.Count).IsActive = True            

Else                                                               

     Items(Items.Count).IsActive = False

End If

Il metodo AddItem() della classe OPCItems ha come parametri l’identificativo dell’item che, come già visto, si può ottenere attraverso il metodo GetItemID() dell’oggetto browser, l’altro parametro è il clientHandle, cioè un identificativo fornito dal client che io scelto come la posizione che occupano nell’elenco Items della schermata principale.

Se il metodo AddItem() và a buon fine, gli items che si è deciso di aggiungere con i  relativi attributi appariranno nell’elenco items della schermata principale

Per selezionare un item cliccare sulla riga corrispondente.

Per modificare un item è necessario selezionare soltanto un item, per altre   

operazioni è possibile effettuare una selezione multipla attraverso i tasti SHIFT e CTRL.

Ogni qualvolta si esegue una lettura dal server vengono aggiornati i seguenti tre attributi della classe OPCItem

· Value : Restituisce l’ultimo valore dell’item letto dal server

· Quality : Restituisce l’ultima quality letta dal server.                                               
      La quality è un indice di quanto sicura sia l’informazione letta dal       

                       server

· TimeStamp : Restituisce l’ultimo timestamp letto dal server.                                            
             Il time stamp rappresenta l’istante di tempo in cui il server    

                             ha prodotto l’informazione 

La lettura di questi tre attributi può avvenire in diversi modi


Lettura sincrona dalla cache


Lettura sincrona dal device


Lettura asincrona dal device


Lettura attraverso l’evento DataChange


Lettura attraverso il Refresh del gruppo a cui appartiene l’item

Per quanto riguarda la scrittura 


Scrittura sincrona da device


Scrittura asincrona da device

Se il client vuole inviati dal server i valori degli items solo quando questi cambiano di valore, allora bisogna utilizzare l’evento DataChange della classe OPCGroup.

Il server per poter eseguire un confronto tra il vecchio valore dell’item e quello nuovo deve naturalmente tenere in memoria il vecchio valore ed è proprio per questo motivo che l’item ed il gruppo in cui si trova l’item devono essere attivi, infatti in questo modo si autorizza il server ad utilizzare la cache per memorizzare il valore dell’item in esame.

L’evento DataChange viene generato quando                                                     
abs(last cached value - current value) > (Deadband/100) * (EU High - EU Low) 

Quindi dipende dalla DeadBand del gruppo, più alto è il suo valore e meno probabilità ci sono che l’evento venga generato, se DeadBand=0 allora l’evento viene generato al minimo cambiamento.

Se si vuole che ogni qualvolta venga generato l’evento DataChange, i cambiamenti degli items vengano visualizzati nell’elenco degli items della schermata principale basta premere nella Barra Pulsanti il pulsante Notifica Data Change e ripremerlo se non si desidera essere avvisati di tali cambiamenti.

Con il metodo SyncRead() della classe OPCItem è possibile effettuare una lettura sincrona dalla cache o dal device.

Quando la lettura è sincrona il client fa la richiesta di lettura di uno o più item e attende sin quando il server non gli fornisce i valori, il vantaggio è che dati acquisiti dal client sono “freschi” nel senso che sono quelli appena letti dal device o dalla cache, lo svantaggio è che si introduco ritardi, infatti ogni richiesta al server per essere eseguita deve attendere che sia conclusa quella precedente.

Per effettuare una lettura sincrona, selezionare nell’elenco item gli items che si desidera leggere, poi nel frame Read Items selezionare Sincrona nella casella combinata, scegliere se voler leggere dalla cache o dal device e premere Read, l’istruzione che permette la lettura sincrona è

CollezioneGruppi(i).group.SyncRead Source, NumItems, serverhandle, Value, Errors,  

                                                                        Qualities, TimeStamp

Bisogna specificare i primi tre parametri                                                                          Source : può contenere le due costanti OPCCache o OPCDevice                            NumItems : numero di items da leggere                                                              Serverhandle : array contenente i serverhandle degli item che si desidera leggere. I serverhandle sono degli identificativi (a cui si accede attraverso l’attributo serverhandle dell’oggetto OPCItem) che il server attribuisce a ciascun item nel momento in cui lo aggiunge ad un gruppo.

Gli altri quattro parametri verranno utilizzati dal metodo per restituire rispettivamente un array dei valori degli item, un array di errori per verificare la buona riuscita su ciascun item, un array delle quality ed un array dei timestamp.

Quando il client chiede al server una richiesta di lettura asincrona non si blocca in attesa che il server effettui tale operazione come nella lettura sincrona, ma è libero di fare altre richieste al server, sarà poi il server che quando ha terminato l’operazione richiesta avvisa il client attraverso l’evento AsyncReadComplete.

Il vantaggio è che non si introducono ritardi, lo svantaggio è che poiché il server avrà una coda di richieste da effettuare, trascorrerà del tempo tra la richiesta del client e l’istante di tempo in cui il server restituisce i dati e quindi i dati non sono “freschi”.

Per effettuare una lettura asincrona (è possibile farla solo sul device), selezionare nell’elenco item gli items che si desidera leggere, poi nel frame Read Items selezionare Asincrona nella casella combinata e premere Read, l’istruzione che permette la lettura asincrona è

CollezioneGruppi(i).group.AsyncRead NumItems, serverhandle, Errors, TransactionID, 

                                                                           CancelID

Bisogna specificare i seguenti parametri                                                                     NumItems : Il numero di item  che devono essere letti                                     Serverhandle : array dei serverhandle degli items forniti dal server                TransactionID : identificativo della transazione

I parametri Errors e CancelID saranno utilizzati dal metodo per restituire rispettivamente un array di errori per verificare il successo delle singole operazioni di lettura degli items e un identificativo generato dal server per permettere al client di cancellare la ‘transaction’.

Con il metodo AsyncRefresh() dell’oggetto OPCGroup viene effettuata una lettura di tutti gli items attivi del gruppo specificato, il quale naturalmente deve essere anche attivo. Per effettuare il refresh di uno o più gruppi, selezionare nell’elenco Gruppi della schermata principale i gruppi desiderati e successivamente nel frame Refresh Groups scegliere se effettuare il refresh da cache o da device e quindi premere il pulsante Refresh, l’istruzione che permette di eseguire il refresh è


‘GridGroups.Row restituisce l’indice della riga selezionata nell’elenco gruppi 
CollezioneGruppi(GridGroups.Row).group.AsyncRefresh Source, TransactionID, CancelID 

Il metodo AsyncRefresh() vuole specificati i parametri                                                         Source  : può contenere le due costanti OPCCache o OPCDevice.                                                        
      Se si effettua il refresh dal device, allora tutti gli item presenti nella  

               Cache subiranno un refresh dal device prima della callback.                              TransactionID : deve essere un valore diverso da zero

Il parametro CancelID sarà utilizzato dal metodo per restituire un identificativo generato dal server per permettere al client di cancellare la ‘transaction’.

Quando il server termina l’operazione genera l’evento DataChange, quindi l’evento DataChange viene generato sia quando si verifica un cambiamento dei valori degli items sia quando viene eseguito il metodo AsyncRefresh(), per distinguerli allora bisogna controllare il valore di TransactionID, se questo è diverso da zero allora vorrà dire che l’evento DataChange è stato generato a causa del metodo AsyncRefresh(), altrimenti vorrà dire che è stato generato a causa di una variazione dei valori degli items.

La scrittura di un item può avvenire solo su device.

Per scrivere un item selezionarlo dall’elenco items della schermata principale, successivamente nel frame Write Item selezionare nella casella combinata se effettuare una scrittura di tipo sincrona o asincrona e immettere il valore nella casella di testo Value, quindi premere Write, l’istruzione che permette la scrittura sincrona è


group.SyncWrite 1, serverhandle, Value, Errors

I primi tre parametri rappresentano rispettivamente il numero di item da leggere, un array contenente i serverhandle degli items (in questo caso è un array con un solo elemento) e un array dei valori da scrivere (anche questo è costituito da un solo elemento). Errors è un array di errori che sarà restituito dal metodo per verificare il successo dell’operazione su ciascun item.

L’istruzione che permette una lettura asincrona è


group.AsyncWrite 1, serverhandle, Value, Errors, TransactionID, CancelID

Rispetto al metodo SyncWrite() bisogna specificare anche un identificativo della transazione TransactionID, inoltre in CancelID sarà restituito un identificativo generato dal server per permettere al client di cancellare la ‘transaction’.

Quando viene creato un gruppo con il metodo Add() le proprietà IsActive, UpdateRate, DeadBand, TimeBias assumeranno i valori DefaultGroupIsActive, DefaultGroupUpdateRate, DefaultGroupDeadBand, DefaultGroupTimeBias dell’oggetto OPCGroups a cui appartiene il gruppo e solo dopo che il gruppo è stato aggiunto è possibile modificare le sue proprietà.

Per modificare le proprietà di default dell’oggetto OPCGroups selezionare dal menù a tendina OPCGroups e quindi Proprietà di Default, apparirà la seguente finestra

In Nome Server è possibile scegliere il server di cui l’oggetto OPCGroups fa parte; all’apertura della finestra nella casella combinata apparirà il nome del server selezionato nel TreeView.

Nelle altre caselle, all’apertura della finestra appariranno i valori correnti dell’oggetto OPCGroups.

Premendo il pulsante Conferma verranno memorizzati i nuovi valori attraverso le seguenti istruzioni

ServerConnessi(NomeServer.ListIndex + 1).server.OpcGroups.DefaultGroupDeadband = DeadBand.Text

ServerConnessi(NomeServer.ListIndex + 1).server.OpcGroups.DefaultGroupTimeBias = TimeBias.Text

ServerConnessi(NomeServer.ListIndex + 1).server.OpcGroups.DefaultGroupUpdateRate = UpdateRate.Text

If CheckIsActive.Value = 1 Then ‘Controlla che la casella Default Is Active sia selezionata

   ServerConnessi(NomeServer.ListIndex + 1).server.OpcGroups.DefaultGroupIsActive = True

Else

   ServerConnessi(NomeServer.ListIndex + 1).server.OpcGroups.DefaultGroupIsActive = False

End If

Quando viene aggiunto un item ad un gruppo le proprietà IsActive e RequestedDataType, assumeranno i valori DefaultIsActive e DefaultRequestedDataType dell’oggetto OPCItems a cui appartiene l’item e solo dopo che l’item è stato aggiunto è possibile modificare le sue proprietà.

Per modificare le proprietà di default dell’oggetto OPCItems selezionare dal menù a tendina OPCItems e quindi Proprietà di Default, apparirà la seguente finestra

Attraverso i campi Server e Gruppo è possibile scegliere il server ed il gruppo di cui l’oggetto OPCItems fa parte; all’apertura della finestra, nella casella combinata Gruppo apparirà il nome del gruppo selezionato nel TreeView.

Nelle altre caselle, all’apertura della finestra appariranno i valori correnti dell’oggetto OPCItems.

Premendo il pulsante Conferma verranno memorizzati i nuovi valori attraverso le seguenti istruzioni

          ‘CodeDataType() è una funzione dichiarata nel Modulo Public che torna il codice del tipo di   

          ‘ dato espresso come stringa

          ServerConnessi(i).server.OpcGroups(j).OpcItems.DefaultRequestedDataType =  

          CodeDataType(ComboRequestedDataType.Text)

          If CheckIsActive.Value = 1 Then   ‘Controlla che la casella Default Is Active si selezionata
             ServerConnessi(i).server.OpcGroups(j).OpcItems.DefaultIsActive = True

          Else

             ServerConnessi(i).server.OpcGroups(j).OpcItems.DefaultIsActive = False

          End If

Per modificare le proprietà di un gruppo, selezionare un gruppo dall’elenco gruppi della schermata principale e premere il pulsante Modifica Gruppo della Barra Pulsanti oppure selezionare OPCGroups dal menu a tendina e quindi Modifica Proprietà, apparirà la seguente finestra

I campi Nome Server e Nome Gruppo mettono in evidenza il gruppo a cui saranno apportate le modifiche.

Una volta scelti i nuovi attributi, premere il pulsante Conferma Modifiche, parte del codice che verrà eseguito è il seguente
If CheckIsActive.Value = 1 Then

   ‘FormElencoItem.GridGroups.Row  torna la riga del gruppo selezionato

   CollezioneGruppi(FormElencoItem.GridGroups.Row).group.IsActive = True

Else

   CollezioneGruppi(FormElencoItem.GridGroups.Row).group.IsActive = False

End If

If CheckIsSubscribed.Value = 1 Then

   CollezioneGruppi(FormElencoItem.GridGroups.Row).group.IsSubscribed = True

Else

   CollezioneGruppi(FormElencoItem.GridGroups.Row).group.IsSubscribed = False

End If

If CollezioneGruppi(FormElencoItem.GridGroups.Row).group.Name <> NuovoNomeGruppo.Text Then

   CollezioneGruppi(FormElencoItem.GridGroups.Row).group.Name = NuovoNomeGruppo.Text

End If

CollezioneGruppi(FormElencoItem.GridGroups.Row).group.DeadBand = DeadBand.Text

CollezioneGruppi(FormElencoItem.GridGroups.Row).group.TimeBias = TimeBias.Text

CollezioneGruppi(FormElencoItem.GridGroups.Row).group.UpdateRate = UpdateRate.Text

Per modificare le proprietà di un item, selezionare un item dall’elenco item della schermata principale e premere il pulsante Modifica Item dalla Barra Pulsanti, oppure selezionare OPCItems dal menu a tendina e quindi Modifica Proprietà, apparirà la seguente finestra

I campi Nome Server, Nome Gruppo e Nome Item mettono in evidenza il gruppo a cui saranno apportate le modifiche, mentre il campo Canonical Data Type rappresenta il tipo di dato originario dell’item nel server e naturalmente è di sola lettura.

Una volta scelti i nuovi attributi, premere il pulsante Conferma Modifiche, parte del codice che verrà eseguito è il seguente

‘ L’istruzione sottostante assegna alla variabile Item di tipo OPCItem il riferimento all’item               ‘ selezionato, per far questo ho utilizzato il metodo GetOPCGroup() per ottenere il riferimento del     ‘ gruppo a cui l’item appartiene ed il metodo GetOPCItem() per ottenere il riferimento allo stesso      ‘ item. Il metodo GetOPCGroup() vuole come parametro il nome del gruppo, che nel nostro caso è       ‘restituito dall’istruzione FormElencoItem.GridItems.TextMatrix(FormElencoItem.GridItems.Row,1 ‘ ), mentre il metodo GetOPCItem() vuole come parametro il serverhandle dell’item, che è restituito  ‘ dall’istruzione ServerHandleItem(FormElencoItem.GridItems.Row) dove serverhandleItem è un     ‘ array contenente tutti serverhandle degli item aggiunti ed è dichiarato nel modulo Public

Set Item = ServerConnessi(i).server.OpcGroups.GetOPCGroup (FormElencoItem.GridItems.  

                                    TextMatrix(FormElencoItem.GridItems.Row, 1)). 

                                    OpcItems.GetOPCItem(ServerHandleItem(FormElencoItem.GridItems.Row))

Item.RequestedDataType = CodeDataType(ComboRequestedDataType.Text)

If CheckIsActive.Value = 1 Then

   Item.IsActive = True

Else

   Item.IsActive = False

End If
Per rimuovere gli item da un gruppo selezionarli dall’elenco item della schermata principale, successivamente selezionare dal menù a tendina OPCItems e quindi Rimuovi Item Selezionati.

L’istruzione che permette la rimozione degli Item è la seguente


CollezioneGruppi(i).group.OpcItems.Remove 1, serverhandle, Errors

Il metodo Remove() ha tre parametri:                                                                    
nel primo bisogna specificare il numero di item che si desidera eliminare  
(in questo caso 1)
il secondo parametro deve essere un array dei    

          serverhandle degli item da eliminare, il terzo parametro sarà utilizzato dal  

          metodo per restituire un array di errori al fine di verificare il successo  

          dell’operazione su ciascun item 

Se il  metodo Remove() ha successo gli item selezionati scompariranno dall’elenco degli item della schermata principale

Per rimuovere i gruppi selezionarli dall’elenco gruppi della schermata principale, successivamente selezionare dal menù a tendina OPCGroups e quindi Rimuovi Gruppi Selezionati.

L’istruzione che permette la rimozione degli Item è la seguente 


ServerConnessi(j).server.OpcGroups.Remove (FormElencoItem.GridGroups.TextMatrix(i, 1))

L’unico parametro del metodo Remove() è il nome del gruppo da eliminare, che in questo caso è restituito dall’istruzione FormElencoItem.GridGroups.TextMatrix(i, 1). Quest’ultima istruzione torna il valore della cella selezionata nell’elenco gruppi della schermata principale.

Se il  metodo Remove() ha successo i gruppi selezionati scompariranno dall’elenco dei gruppi della schermata principale.

Per visualizzare le proprietà di un server a cui ci si è connessi, selezionare OPCServer dal menù a tendina e quindi Proprietà, apparirà la seguente finestra.


Attraverso il campo Server Name è possibile scegliere di quale dei server connessi visualizzare gli attributi.

All’apertura della finestra la casella combinata Server Name conterrà il nome del server selezionato nel TreeView.

Le istruzioni che permettono la visualizzazione degli attributi del server selezionato sono le seguenti

StartTime.Caption = ServerConnessi(i).server.StartTime

CurrentTime.Caption = ServerConnessi(i).server.CurrentTime

LastUpdateTime.Caption = ServerConnessi(i).server.LastUpdateTime

MajorVersion.Caption = ServerConnessi(i).server.MajorVersion

MinorVersion.Caption = ServerConnessi(i).server.MinorVersion

BuildNumber.Caption = ServerConnessi(i).server.BuildNumber

VendorInfo.Caption = ServerConnessi(i).server.VendorInfo

ServerState.Caption = ServerConnessi(i).server.ServerState

LocalID.Caption = ServerConnessi(i).server.LocaleID

BandWidth.Caption = ServerConnessi(i).server.BandWidth

ServerNode.Caption = ServerConnessi(i).server.ServerNode

Per disconnettersi da un server basta selezionare dal menù a tendina OPCServers e quindi Disconnetti, apparirà la seguente finestra


All’apertura della finestra nella casella combinata Nome Server apparirà il nome del server selezionato nel TreeView, successivamente e possibile scegliere un qualsiasi server da cui ci si vuole disconnettere e premere il pulsante Disconnetti, l’istruzione che permette la disconnessione è la seguente


ServerConnessi(i).server.Disconnect 

Come si può notare il metodo Disconnect() non richiede alcun parametro, l’importante è che l’oggetto server sia quello relativo al server da eliminare.

L’ultima cosa che resta da commentare è l’evento ServerShutDown della classe OPCServer.

Questo evento viene generato nel momento in cui un server si disconnette per una qualche ragione.

Nel mio progetto le istruzioni da eseguire quando viene generato tale evento si trovano nel modulo di classe OPCServers che ricordo è una collezione di oggetti OPCServer

Private Sub Server_ServerShutDown(ByVal Reason As String)

        Istruzioni

End Sub

Il Parametro Reason torna una stringa contenente il motivo per cui il server in questione si è disconnesso.

